-module(euler11).
-author("Jeff Zellner <jeff.zellner@gmail.com>").
-export([go/0]).

%% Problem 11:
%% What is the greatest product of four numbers on 
%% the same straight line in the 20 by 20 grid?

%% my understanding of how to do this problem in erlang is cribbed from
%%      http://www.smoil.com/2009/05/erlang-and-project-euler-problem-11/

%% start solving for a given direction

max(Matrix, Length, Direction) when 
    Direction =:= horizontal; Direction =:= vertical; Direction =:= diagonal_right ->

    Dimension = {length(Matrix), length(lists:nth(1,Matrix))},
    max(Matrix, Length, Dimension, {1, 1}, 0, Direction);

max(Matrix, Length, Direction) when 
    Direction =:= diagonal_left ->

    Dimension = {length(Matrix), length(lists:nth(1, Matrix))},
    max(Matrix, Length, Dimension, {1, Length}, 0, Direction).
    
%% return last row result

max(_, _, {Rows, _}, {Row, _}, Result, Direction) when
    Row > Rows, Direction =:= horizontal ->

    Result;

max(_, Length, {Rows, _}, {Row, _}, Result, Direction) when
    Row+Length-1 > Rows, (Direction =:= vertical) or
    (Direction =:= diagonal_left) or (Direction =:= diagonal_right) ->

    Result;

%% go down a row

max(Matrix, Length, {_, Cols} = Dimension, {Row, Col}, Result, Direction) when
    Col+Length-1 > Cols, (Direction =:= horizontal) or
    (Direction =:= diagonal_right) ->

    max(Matrix, Length, Dimension, {Row+1, 1}, Result, Direction);

max(Matrix, Length, {_, Cols} = Dimension, {Row, Col}, Result, Direction) when
    Col > Cols, Direction =:= diagonal_left ->

    max(Matrix, Length, Dimension, {Row + 1, Length}, Result, Direction);

max(Matrix, Length, {_, Cols} = Dimension, {Row, Col}, Result, Direction) when
    Col > Cols, Direction =:= vertical ->

    max(Matrix, Length, Dimension, {Row + 1, 1}, Result, Direction);

%% go over one column and update result if needed

max(Matrix, Length, Dimension, {Row, Col} = Position, Result, Direction) ->
    Product = line_product(Position, Length, Matrix, Direction),
    case Product > Result of
        true -> max(Matrix, Length, Dimension, {Row, Col + 1}, Product, Direction);
        false -> max(Matrix, Length, Dimension, {Row, Col + 1}, Result, Direction)
    end.

%% compute product of a line

line_product(_, 0, _, _) -> 
    1;
line_product({Row, Col} = Start, Length, Matrix, Direction) ->
    case Direction of
        horizontal -> 
            get_xy(Start, Matrix) * 
            line_product({Row, Col + 1}, Length - 1, Matrix, Direction);
        vertical ->
            get_xy(Start, Matrix) *
            line_product({Row + 1, Col}, Length - 1, Matrix, Direction);
        diagonal_right ->
            get_xy(Start, Matrix) *
            line_product({Row + 1, Col + 1}, Length - 1, Matrix, Direction);
        diagonal_left ->
            get_xy(Start, Matrix) *
            line_product({Row + 1, Col - 1}, Length - 1, Matrix, Direction)
    end.

%% get value of cell

get_xy({Row, Col}, Matrix) ->
    lists:nth(Col, lists:nth(Row, Matrix)).

%% run problem

go() ->
    ProblemMatrix = [[08,02,22,97,38,15,00,40,00,75,04,05,07,78,52,12,50,77,91,08],
                     [49,49,99,40,17,81,18,57,60,87,17,40,98,43,69,48,04,56,62,00],
                     [81,49,31,73,55,79,14,29,93,71,40,67,53,88,30,03,49,13,36,65],
                     [52,70,95,23,04,60,11,42,69,24,68,56,01,32,56,71,37,02,36,91],
                     [22,31,16,71,51,67,63,89,41,92,36,54,22,40,40,28,66,33,13,80],
                     [24,47,32,60,99,03,45,02,44,75,33,53,78,36,84,20,35,17,12,50],
                     [32,98,81,28,64,23,67,10,26,38,40,67,59,54,70,66,18,38,64,70],
                     [67,26,20,68,02,62,12,20,95,63,94,39,63,08,40,91,66,49,94,21],
                     [24,55,58,05,66,73,99,26,97,17,78,78,96,83,14,88,34,89,63,72],
                     [21,36,23,09,75,00,76,44,20,45,35,14,00,61,33,97,34,31,33,95],
                     [78,17,53,28,22,75,31,67,15,94,03,80,04,62,16,14,09,53,56,92],
                     [16,39,05,42,96,35,31,47,55,58,88,24,00,17,54,24,36,29,85,57],
                     [86,56,00,48,35,71,89,07,05,44,44,37,44,60,21,58,51,54,17,58],
                     [19,80,81,68,05,94,47,69,28,73,92,13,86,52,17,77,04,89,55,40],
                     [04,52,08,83,97,35,99,16,07,97,57,32,16,26,26,79,33,27,98,66],
                     [88,36,68,87,57,62,20,72,03,46,33,67,46,55,12,32,63,93,53,69],
                     [04,42,16,73,38,25,39,11,24,94,72,18,08,46,29,32,40,62,76,36],
                     [20,69,36,41,72,30,23,88,34,62,99,69,82,67,59,85,74,04,36,16],
                     [20,73,35,29,78,31,90,01,74,31,49,71,48,86,81,16,23,57,05,54],
                     [01,70,54,71,83,51,54,69,16,92,33,48,61,43,52,01,89,19,67,48]],
    lists:max([max(ProblemMatrix, 4, horizontal),
               max(ProblemMatrix, 4, vertical),
               max(ProblemMatrix, 4, diagonal_left),
               max(ProblemMatrix, 4, diagonal_right)]).

